

CSCE-608 Database Systems

Spring 2025

Instructor: Dr. Jianer Chen

Office: PETR 428

Phone: 845-4259

Email: chen@cse.tamu.edu

Office Hours: MW 1:30 pm–3:00 pm

TA: Shahin John John Stella

Office: PETR 445

Phone: (979) 575-5523

Email: ivin-98@tamu.edu

Office Hours: TR 11:00 am–12:00 pm

Tiny-SQL Grammar

I. Lexical Rules

```
letter ::= a | b | c | d | e | f | g | h | i | j | k | l | m | n | o | p | q | r
        | s | t | u | v | w | x | y | z
digit  ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9
integer ::= digit+
comp-op ::= < | > | =
table-name ::= letter[digit | letter]*
attribute-name ::= letter[digit | letter]*
column-name ::= [table-name.]attribute-name
literal ::= "whatever-except-("-and-")"
```

II. Syntax Rules

statement ::= create-table-statement | drop-table-statement | select-statement
| delete-statement | insert-statement

create-table-statement ::= CREATE TABLE table-name(attribute-type-list)

data-type ::= INT | STR20

attribute-type-list ::= attribute-name data-type
| attribute-name data-type, attribute-type-list

drop-table-statement ::= DROP TABLE table-name

select-statement ::= SELECT [DISTINCT] select-list
FROM table-list
[WHERE search-condition]
[ORDER BY column-name]

select-list ::= * | select-sublist

select-sublist ::= column-name | column-name, select-sublist

table-list ::= table-name | table-name, table-list

delete-statement ::= DELETE FROM table-name [WHERE search-condition]

insert-statement ::= INSERT INTO table-name(attribute-list) insert-tuples

insert-tuples ::= VALUES (value-list) | select-statement

attribute-list ::= attribute-name | attribute-name, attribute-list

value ::= literal | integer | NULL

value-list ::= value | value, value-list

search-condition ::= boolean-term | boolean-term OR search-condition

boolean-term ::= boolean-factor | boolean-factor AND boolean-term

boolean-factor ::= expression comp-op expression

expression ::= term | (term + term) | (term - term) | (term * term)

term ::= column-name | literal | integer